In-class Demonstration: Inverses

By Russell Blyth

```
> restart: with(LinearAlgebra):
```

We start with a random 3 x 3 matrix.

```
> A:= RandomMatrix(3,3,generator=rand(-3..3));
```

$$A := \begin{bmatrix} 1 & 3 & 3 \\ -2 & 1 & 2 \\ -2 & -1 & 0 \end{bmatrix} \quad (1)$$

Our task is to find the inverse of the matrix A.
The equation Y = AX computes Y given X. We need to compute X given Y. Define a general Y, and solve for X:

```
> Y := Vector(3,<y1,y2,y3>);
  AugA := <A|Y>;
  RedA := ReducedRowEchelonForm(AugA);
```

$$Y := \begin{bmatrix} y1 \\ y2 \\ y3 \end{bmatrix}$$

$$AugA := \begin{bmatrix} 1 & 3 & 3 & y1 \\ -2 & 1 & 2 & y2 \\ -2 & -1 & 0 & y3 \end{bmatrix}$$

$$RedA := \begin{bmatrix} 1 & 0 & 0 & y1 + \frac{3}{2}\,y3 - \frac{3}{2}\,y2 \\ 0 & 1 & 0 & 3\,y2 - 2\,y1 - 4\,y3 \\ 0 & 0 & 1 & \frac{7}{2}\,y3 + 2\,y1 - \frac{5}{2}\,y2 \end{bmatrix} \quad (2)$$

Next we extract the expression for X in terms of Y

```
> Xvec := DeleteColumn(RedA,1..3);
```

$$Xvec := \begin{bmatrix} y1 + \frac{3}{2}\,y3 - \frac{3}{2}\,y2 \\ 3\,y2 - 2\,y1 - 4\,y3 \\ \frac{7}{2}\,y3 + 2\,y1 - \frac{5}{2}\,y2 \end{bmatrix} \quad (3)$$

```
> y1vec := eval(Xvec,{y1=1,y2=0,y3=0});
  y2vec := eval(Xvec,{y1=0,y2=1,y3=0});
  y3vec := eval(Xvec,{y1=0,y2=0,y3=1});
```

$$y1vec := \begin{bmatrix} 1 \\ -2 \\ 2 \end{bmatrix}$$

$$y2vec := \begin{bmatrix} \frac{-3}{2} \\ 3 \\ \frac{-5}{2} \end{bmatrix}$$

$$y3vec := \begin{bmatrix} \frac{3}{2} \\ -4 \\ \frac{7}{2} \end{bmatrix} \qquad (4)$$

Thus Xvec = y1*y1vec + y2*y2vec + y3*y3vec. If we make a matrix with y1vec, y2vec, y3vec as the columns, then y1, y2 and y3 are the coefficients of the linear combination giving X. We write this as BY, where

```
> B := <y1vec | y2vec | y3vec>;
```

$$B := \begin{bmatrix} 1 & \frac{-3}{2} & \frac{3}{2} \\ -2 & 3 & -4 \\ 2 & \frac{-5}{2} & \frac{7}{2} \end{bmatrix} \qquad (5)$$

Check:

```
> B.Y;
```

$$\begin{bmatrix} y1 + \frac{3}{2}\, y3 - \frac{3}{2}\, y2 \\ 3\, y2 - 2\, y1 - 4\, y3 \\ \frac{7}{2}\, y3 + 2\, y1 - \frac{5}{2}\, y2 \end{bmatrix} \qquad (6)$$

This is just the transpose of Xvec. Thus X = BY. B is the inverse of A!

Check the products AB and BA:

```
> A.B;
  B.A;
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \tag{7}$$

Note the first column of B arises from solving AX = $[1, 0, 0]^t$; the second column from solving AX = $[0, 1, 0]^t$ and the third column from solving AX = $[0, 0, 1]^t$. These three systems may be solved simultaneously, by finding the reduced row echelon form of the matrix [A,I]

```
> I3 := IdentityMatrix(3);
  BigA := <A | I3>;
  RedBigA := ReducedRowEchelonForm(BigA);
```

$$I3 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$BigA := \begin{bmatrix} 1 & 3 & 3 & 1 & 0 & 0 \\ -2 & 1 & 2 & 0 & 1 & 0 \\ -2 & -1 & 0 & 0 & 0 & 1 \end{bmatrix}$$

$$RedBigA := \begin{bmatrix} 1 & 0 & 0 & 1 & \frac{-3}{2} & \frac{3}{2} \\ 0 & 1 & 0 & -2 & 3 & -4 \\ 0 & 0 & 1 & 2 & \frac{-5}{2} & \frac{7}{2} \end{bmatrix} \tag{8}$$

Note the final three columns are precisely the inverse matrix B.

Maple computes matrix inverses directly:

```
> A^(-1);
```

$$\begin{bmatrix} 1 & \frac{-3}{2} & \frac{3}{2} \\ -2 & 3 & -4 \\ 2 & \frac{-5}{2} & \frac{7}{2} \end{bmatrix} \tag{9}$$

```
>
```